

August 2008

Spreadsheet Implementations for Solving Power-Flow Problems

Mark A. Lau

Universidad del Turabo, mlau@suagm.edu

Sastry Kuruganty

Universidad del Turabo, ut_skurugant@suagm.edu

Follow this and additional works at: <http://epublications.bond.edu.au/ejsie>



This work is licensed under a [Creative Commons Attribution-Noncommercial-No Derivative Works 4.0 License](https://creativecommons.org/licenses/by-nc-nd/4.0/).

Recommended Citation

Lau, Mark A. and Kuruganty, Sastry (2008) Spreadsheet Implementations for Solving Power-Flow Problems, *Spreadsheets in Education (eJSiE)*: Vol. 3: Iss. 1, Article 3.

Available at: <http://epublications.bond.edu.au/ejsie/vol3/iss1/3>

This In the Classroom Article is brought to you by the Bond Business School at [epublications@bond](mailto:epublications@bond.edu.au). It has been accepted for inclusion in Spreadsheets in Education (eJSiE) by an authorized administrator of [epublications@bond](mailto:epublications@bond.edu.au). For more information, please contact [Bond University's Repository Coordinator](mailto:epublications@bond.edu.au).

Spreadsheet Implementations for Solving Power-Flow Problems

Abstract

The solution to the power-flow problem is of fundamental importance in power system analysis and design. Transient stability studies and fault analysis in power systems demand solutions to a power-flow problem as a first step in the analysis. Although commercial software such as PSS/E (Power System Simulator for Engineers) or PowerWorld can be used for power flow calculations, such specialized programs may not be widely available in many universities or colleges. Spreadsheets, on the other hand, provide an economic alternative for the implementation of the numerical algorithms encountered in power flows.

Keywords

power flows, Gauss-Seidel method, Newton-Raphson method, power systems

Distribution License



This work is licensed under a [Creative Commons Attribution-Noncommercial-No Derivative Works 4.0 License](https://creativecommons.org/licenses/by-nc-nd/4.0/).

Cover Page Footnote

The authors express their appreciation to the Universidad del Turabo where this work was conducted.

Spreadsheet Implementations for Solving Power-Flow Problems

Mark Lau

*Department of Electrical and Computer Engineering
Universidad del Turabo
mlau@suagm.edu*

Sastry Kuruganty

*Department of Electrical and Computer Engineering
Universidad del Turabo
ut_skurugant@suagm.edu*

Abstract

The solution to the power-flow problem is of fundamental importance in power system analysis and design. Transient stability studies and fault analysis in power systems demand solutions to a power-flow problem as a first step in the analysis. Although commercial software such as PSS/E (Power System Simulator for Engineers) or PowerWorld can be used for power flow calculations, such specialized programs may not be widely available in many universities or colleges. Spreadsheets, on the other hand, provide an economic alternative for the implementation of the numerical algorithms encountered in power flows.

Keywords: power flows, Gauss-Seidel method, Newton-Raphson method, power systems.

1. Introduction

The solution to the power-flow problem is of fundamental importance in power system analysis and design. In transient stability studies and fault analysis, solutions to a power-flow problem constitute a necessary initial step in such analyses.

The objective of the power-flow problem is to calculate the voltage magnitudes and phase angles at each bus or node in a given power system. Calculations are performed under the assumption of balanced three-phase steady-state conditions. In addition to voltages and angles, real and reactive power flows in equipment such as transformers and transmission lines can be also obtained from these calculations.

The topics in this paper follow the treatment found in standard reference material on power system analysis and design [2–4]. In particular, two numerical methods, namely, the Gauss-Seidel and the Newton-Raphson methods are used to determine the power flows in a small-scale power system. The examples are simple enough so that readers can replicate hand calculations and reproduce the spreadsheet implementations. The application of spreadsheets for solving power flows and other related problems has been reported in the literature [1, 5, 6]. The emphasis of this paper is on the educational value of spreadsheets in the analysis of power systems.

The paper is organized as follows: Section 2 presents the procedure for constructing the bus admittance matrix of a power network; Section 3 presents the description of the

general power-flow problem; Section 4 discusses the solution to the power-flow problem using the Gauss-Seidel method and presents a spreadsheet implementation of this technique; Section 5 provides the solution to the power-flow problem using the Newton-Raphson method along with its spreadsheet implementation; and Section 6 presents concluding remarks.

2. The bus admittance matrix of a power network

The bus admittance matrix offers a convenient representation of the topology of a power network. A power network contains various components, namely, bus bars, generators, transformers, transmission lines, loads, and so on. The network is described by the bus (or node) voltages and the equivalent admittances of components connected between buses or between buses and ground. In the following, we provide a simple constructive approach to determine the bus admittance matrix of a power network. This network representation is referred to as the *primitive representation*. The relation between the injected node currents and the (line-to-neutral) node voltages is given by

$$\mathbf{I} = \mathbf{Y}_{\text{bus}}\mathbf{V} \quad (1)$$

where \mathbf{I} is the vector of injected node currents, \mathbf{Y}_{bus} is the bus admittance matrix, and \mathbf{V} is the vector of node voltages.

Each component element of the interconnected network is referred to as a *branch*. The branch is connected between two nodes of the network, or between a node and the reference node (ground).

Construction of \mathbf{Y}_{bus} by inspection

- The \mathbf{Y}_{bus} matrix is symmetric.
- The diagonal element Y_{ii} of \mathbf{Y}_{bus} is the self-admittance and is equal to the sum of the primitive admittances of all the components connected to the i th node.
- The ij th off-diagonal element Y_{ij} of \mathbf{Y}_{bus} is equal to the negative of the primitive admittance of all components connected between nodes i and j .

As an example, we consider the simple power system depicted in Figure 1. The system consists of three buses: a synchronous generator connected to bus 1, a load connected to bus 2, a voltage-regulated generator connected to bus 3, and transmission lines between buses 1 and 2, 2 and 3, and 1 and 3. The parameters for the network components are given in Tables 1 and 2. All parameters are expressed in per unit (p.u.) of a reference base.




Figure 1: Three-bus power system.

Table 1: Bus input data.

Type	Bus i	Bus voltage		Complex power (p.u.)			
		$ V_i $ (p.u.)	δ_i (deg.)	P_{gi}	Q_{gi}	P_{di}	Q_{di}
Slack	1	1.0	0°	–	–	0.0	0.0
Load	2	–	–	0.0	0.0	2.0	0.5
Constant voltage	3	1.0	–	1.0	–	0.0	0.0

Table 2: Transmission line parameters.

Line bus i to bus k	Impedance Z_{ik} (p.u.)
1–2	$j0.1$
2–3	$j0.2$
1–3	$j0.4$

To determine the \mathbf{Y}_{bus} matrix for the given system, we calculate the primitive admittance of each branch in per unit:

$$y_{12} = \frac{1}{j0.1} = -j10,$$

$$y_{23} = \frac{1}{j0.2} = -j5,$$

$$y_{13} = \frac{1}{j0.4} = -j2.5.$$

With the primitive admittances just calculated, we determine the elements of \mathbf{Y}_{bus} :

$$\begin{aligned} Y_{11} &= y_{12} + y_{13} = -j10 - j2.5 = -j12.5, \\ Y_{22} &= y_{12} + y_{23} = -j10 - j5 = -j15, \\ Y_{33} &= y_{23} + y_{13} = -j5 - j2.5 = -j7.5, \\ Y_{12} &= Y_{21} = -y_{12} = j10, \\ Y_{23} &= Y_{32} = -y_{23} = j5, \\ Y_{13} &= Y_{31} = -y_{13} = j2.5. \end{aligned}$$

Therefore, the bus admittance matrix for the network of Figure 1 is given by

$$\mathbf{Y}_{\text{bus}} = \begin{bmatrix} -j12.5 & j10 & j2.5 \\ j10 & -j15 & j5 \\ j2.5 & j5 & -j7.5 \end{bmatrix}. \quad (2)$$

3. The power-flow problem

The power-flow problem discussed in this section will be presented in terms of the \mathbf{Y}_{bus} matrix whose elements are of the form

$$Y_{ik} = |Y_{ik}| e^{j\theta_{ik}} = |Y_{ik}| \cos \theta_{ik} + j |Y_{ik}| \sin \theta_{ik} = G_{ik} + jB_{ik} \quad (3)$$

for $i, k = 1, 2, \dots, N$. Let the voltage at bus i be denoted by

$$V_i = |V_i| e^{j\delta_i} = |V_i| (\cos \delta_i + j \sin \delta_i) \quad (4)$$

for $i = 1, 2, \dots, N$.

As indicated in Figure 2, the *net* current injected into the network at bus i in terms of the elements Y_{ik} of the \mathbf{Y}_{bus} is determined by

$$I_i = Y_{i1}V_1 + Y_{i2}V_2 + \dots + Y_{iN}V_N = \sum_{k=1}^N Y_{ik}V_k, \quad i = 1, 2, \dots, N. \quad (5)$$

Let P_i and Q_i denote the *net* real and reactive power entering the network at bus i . Then the complex conjugate of the power injected at bus i is

$$P_i - jQ_i = V_i^* \sum_{k=1}^N Y_{ik}V_k = \sum_{k=1}^N |Y_{ik}V_iV_k| e^{j(\theta_{ik} + \delta_k - \delta_i)}. \quad (6)$$

From the preceding equation we obtain the following form of the *power-flow equations*:

$$P_i = \sum_{k=1}^N |Y_{ik}V_iV_k| \cos(\theta_{ik} + \delta_k - \delta_i), \quad (7)$$

$$Q_i = -\sum_{k=1}^N |Y_{ik} V_i V_k| \sin(\theta_{ik} + \delta_k - \delta_i). \quad (8)$$

Now let P_{gi} denote the scheduled power being generated at bus i and P_{di} denote the scheduled power demand of the load at that bus. Then the *net scheduled power* being injected into the network at bus i is

$$P_{i,\text{sch}} = P_{gi} - P_{di}. \quad (9)$$

We define the mismatch ΔP_i as the difference between the net scheduled power $P_{i,\text{sch}}$ and the calculated value of P_i as given by Equation (7). That is,

$$\Delta P_i = P_{i,\text{sch}} - P_i = (P_{gi} - P_{di}) - P_i. \quad (10)$$

Likewise, we may define the mismatch ΔQ_i for the reactive power as

$$\Delta Q_i = Q_{i,\text{sch}} - Q_i = (Q_{gi} - Q_{di}) - Q_i. \quad (11)$$

When the mismatches ΔP_i and ΔQ_i are zero at bus i , we have the *power-balance equations*

$$P_i - P_{i,\text{sch}} = P_i - (P_{gi} - P_{di}) = 0, \quad (12)$$

$$Q_i - Q_{i,\text{sch}} = Q_i - (Q_{gi} - Q_{di}) = 0. \quad (13)$$

The power-flow problem entails the computations of P_i and Q_i for values of the unknown bus voltages which cause the mismatches ΔP_i and ΔQ_i to be equal to zero at each bus.

At each bus i two of the four quantities δ_i , $|V_i|$, P_i , and Q_i are specified and the remaining two are calculated. For convenience bus 1 is designated as the *slack bus* and the voltage angle of the slack bus serves as reference for the angles of all other bus voltages. The usual practice is to set $\delta_1 = 0^\circ$.

4. Power-flow solution by the Gauss-Seidel method

4.1 The Gauss-Seidel method

In practice power-flow solutions are obtained by iterative methods. We consider the three-bus system shown in Figure 1 and derive the general equations for the iterations.

With the slack bus designated as number 1, we start computations at bus 2. If $P_{2,\text{sch}}$ and $Q_{2,\text{sch}}$ are the scheduled real and reactive power, respectively, entering the network at bus 2, we have

$$P_{2,\text{sch}} - jQ_{2,\text{sch}} = V_2^* (Y_{21} V_1 + Y_{22} V_2 + Y_{23} V_3)$$

and solving for the voltage at bus 2

$$V_2 = \frac{1}{Y_{22}} \left[\frac{P_{2,\text{sch}} - jQ_{2,\text{sch}}}{V_2^*} - (Y_{21}V_1 + Y_{23}V_3) \right].$$

Similarly, at load bus 3 we have

$$V_3 = \frac{1}{Y_{33}} \left[\frac{P_{3,\text{sch}} - jQ_{3,\text{sch}}}{V_3^*} - (Y_{31}V_1 + Y_{32}V_2) \right].$$

The iterations are based on the scheduled real and reactive power at buses 2 and 3, the scheduled slack bus voltage $V_1 = |V_1|e^{j\delta_1}$, and initial voltage estimates $V_2^{(0)}$ and $V_3^{(0)}$ at the other buses. The corrected voltage $V_2^{(1)}$ is calculated from

$$V_2^{(1)} = \frac{1}{Y_{22}} \left[\frac{P_{2,\text{sch}} - jQ_{2,\text{sch}}}{V_2^{(0)*}} - (Y_{21}V_1 + Y_{23}V_3^{(0)}) \right].$$

The corrected voltage $V_2^{(1)}$ is then used to calculate the value of $V_3^{(1)}$

$$V_3^{(1)} = \frac{1}{Y_{33}} \left[\frac{P_{3,\text{sch}} - jQ_{3,\text{sch}}}{V_3^{(0)*}} - (Y_{31}V_1 + Y_{32}V_2^{(1)}) \right].$$

The procedure is repeated until the amount of correction in voltage at every bus is less than some predetermined precision index. This process of solving the power-flow equations is known as the *Gauss-Seidel iterative method*.

For an N -bus system the general equation for the calculated voltage at any bus i where P and Q are scheduled is given by

$$V_i^{(m)} = \frac{1}{Y_{ii}} \left[\frac{P_{i,\text{sch}} - jQ_{i,\text{sch}}}{V_i^{(m-1)*}} - \sum_{k=1}^{i-1} Y_{ik} V_k^{(m)} - \sum_{k=i+1}^N Y_{ik} V_k^{(m-1)} \right]. \quad (14)$$

In the following we show the calculations for the first iteration of the Gauss-Seidel method for the three-bus system of Figure 1. The bus admittance matrix \mathbf{Y}_{bus} for this system is given by Equation (2). The input data and unknowns at each bus can be inferred from Table 1 and are summarized in Table 3.

Table 3: Input data and unknowns for the power system of Figure 1.

Type	Bus i	Input data	Unknowns
Slack	1	$ V_1 = 1.0 \text{ p.u.}, \delta_1 = 0^\circ$	$P_{\text{sch},1} = P_{g1} - P_{d1} = P_{g1},$ $Q_{\text{sch},1} = Q_{g1} - Q_{d1} = Q_{g1}$
Load	2	$P_{\text{sch},2} = P_{g2} - P_{d2} = -2.0 \text{ p.u.},$ $Q_{\text{sch},2} = Q_{g2} - Q_{d2} = -0.5 \text{ p.u.}$	$ V_2 , \delta_2$
Constant voltage	3	$ V_3 = 1.0 \text{ p.u.},$ $P_{\text{sch},3} = P_{g3} - P_{d3} = 1.0 \text{ p.u.}$	$\delta_3,$ $Q_{\text{sch},3} = Q_{g3} - Q_{d3} = Q_{g3}$

We assume initial guesses for the voltages $V_2^{(0)} = 1.0e^{j0^\circ} = 1.0$ and $V_3^{(0)} = 1.0e^{j0^\circ} = 1.0$ at buses 2 and 3, respectively. Using Equation (14) we determine an improved value for V_2 :

$$V_2^{(1)} = \frac{1}{-j15} \left[\frac{-2 + j0.5}{1.0} - (j10)(1.0) - (j5)(1.0) \right] = 0.9758e^{-j7.853^\circ}.$$

With $V_2^{(1)}$ at hand, we then proceed to find $Q_3^{(1)}$ and $V_3^{(1)}$. Since bus 3 is a voltage-controlled bus, we can either use Equation (6) or (8) to compute the reactive power at that bus. In the spreadsheet implementation of the Gauss-Seidel method discussed in the next section, we use Equation (6) to compute both the real and reactive power at the buses of interest. We observe that the real power is the real part of the right-hand side of Equation (6), while the reactive power is given by the negative of the imaginary part of the right-hand side of that equation. Here we show how to use Equation (6) to compute the reactive power at bus 3:

$$\begin{aligned} Q_3^{(1)} &= -\text{Im} \left\{ 1.0e^{-j0^\circ} \left[(j2.5)(1.0e^{j0^\circ}) + (j5)(0.9758e^{-j7.853^\circ}) + (-j7.5)(1.0e^{j0^\circ}) \right] \right\} \\ &= 0.1667 \text{ p.u.} \end{aligned}$$

With the preceding value of reactive power, we use Equation (14) once again to calculate an improved value of V_3 :

$$V_3^{(1)} = \frac{1}{-j7.5} \left[\frac{1 - j0.1667}{1.0} - (j2.5)(1.0) - (j5)(0.9758e^{-j7.853^\circ}) \right] = 1.0010e^{j2.545^\circ}.$$

As indicated in Table 3, $|V_3| = 1$ at bus 3 and, therefore, we set $V_3^{(1)} = 1.0e^{j2.545^\circ}$ for next iteration. This concludes the first iteration of the Gauss-Seidel method. The calculations are repeated in a similar manner with updated values until convergence can be discerned. We next show how to implement the above calculations using a spreadsheet.

4.2 Spreadsheet implementation of the Gauss-Seidel method

We begin by inputting the entries that form the bus admittance matrix given in Equation (2). Also, we input the known data shown in Table 3. This step is shown in the screen capture of Figure 2. The actual Microsoft Excel commands used to generate the input data is presented in Figure 3.

SPREADSHEET IMPLEMENTATIONS FOR SOLVING POWER-FLOW PROBLEMS

	A	B	C	D	E	F
1	Bus admittance matrix Ybus					
2	Bus	1	2	3		
3	1	-12.5j	10j	2.5j		
4	2	10j	-15j	5j		
5	3	2.5j	5j	-7.5j		
6						
7	Input data					
8	Type	Bus	V_i (p.u.)	d_i (deg)	P_i (p.u.)	Q_i (p.u.)
9	Swing	1	1	0		
10	Load	2			-2	-0.5
11	Const V	3	1		1	

Figure 2: Bus admittance matrix and input data for the power system of Figure 1.

	A	B	C	D	E	F
1	Bus admittance matrix Ybus					
2	Bus	1	2	3		
3	1	=COMPLEX(0,-12.5,"j")	=COMPLEX(0,10,"j")	=COMPLEX(0,2.5,"j")		
4	2	=COMPLEX(0,10,"j")	=COMPLEX(0,-15,"j")	=COMPLEX(0,5,"j")		
5	3	=COMPLEX(0,2.5,"j")	=COMPLEX(0,5,"j")	=COMPLEX(0,-7.5,"j")		
6						
7	Input data					
8	Type	Bus	V_i (p.u.)	d_i (deg)	P_i (p.u.)	Q_i (p.u.)
9	Swing	1	=COMPLEX(1,0,"j")	=180*IMARGUMENT(C9)/PI()		
10	Load	2			-2	-0.5
11	Const V	3	1		1	

Figure 3: Microsoft Excel commands used to generate the input data.

The calculations of the Gauss-Seidel method are shown in Figure 4 as the cell range H1:T11 in the spreadsheet. The entries in row 3 extending from cell H3 to T3 correspond to initial estimates for the numerical algorithm. The basic formulas are found in row 4 and extend from H4 to T4. The formulas in row 4 are copied to the rows below as many times as needed until convergence is reached. We note that some columns were omitted in Figure 4 for clarity; these columns simply contain intermediate calculations. The calculations reveal that convergence is reached after 8 iterations for a precision index of 10^{-5} for the bus voltages and 10^{-4} for the angles. A final calculations section is also shown in Figure 4; the section covers the cell range V1:AC3, but columns V and W were left out as these columns contain intermediate calculations. The solution to the power-flow problem at hand is summarized in Table 4.

To conclude this section, we provide in Table 5 all the formulas used in the implementation of the Gauss-Seidel method. Formulas corresponding to hidden columns are also included for completeness.

	H	L	M	P	S	T
1	Power-flow solutions by Gauss-Seidel method					
2	Iteration m	 V 2 	d 2	Q 3	 V 3 	d 3
3	0	1.0000	0.000		1.0000	0.000
4	1	0.9758	-7.853	0.1667	1.0000	2.545
5	2	0.9542	-6.976	0.2971	1.0000	3.308
6	3	0.9548	-6.931	0.3063	1.0000	3.363
7	4	0.9549	-6.909	0.3066	1.0000	3.378
8	5	0.9549	-6.904	0.3066	1.0000	3.382
9	6	0.9549	-6.903	0.3066	1.0000	3.383
10	7	0.9549	-6.902	0.3066	1.0000	3.383
11	8	0.9549	-6.902	0.3066	1.0000	3.383

	X	Y	Z	AA	AB	AC
1	Final calculations					
2	P 2	Q 2	P 3	Q 3	P 1	Q 1
3	-2.0000	-0.5000	1.0000	0.3066	1.0000	0.1935

Figure 4: Gauss-Seidel iterations showing relevant quantities in the power system of Figure 1.

Table 4: Steady-state power flows and bus voltages for the system in Figure 1. Except for the angles, all other quantities are expressed in per unit (p.u.).

Bus i	Input data (given)	Unknowns (calculated)
1	$ V_1 = 1.0, \delta_1 = 0^\circ$	$P_{sch,1} = 1.0, Q_{sch,1} = 0.1935$
2	$P_{sch,2} = -2.0, Q_{sch,2} = -0.5$	$ V_2 = 0.9549, \delta_2 = -6.902^\circ$
3	$ V_3 = 1.0, P_{sch,3} = 1.0$	$\delta_3 = 3.383^\circ, Q_{sch,3} = 0.3066$

Table 5: Microsoft Excel formulas for implementing the Gauss-Seidel method.

Cell	Formula	Comments
H3:H11	List of numbers from 0 to 8.	Iteration number.
H2:T2	Labels for various quantities.	Gauss-Seidel section.
I3, J3	Blank cells.	No calculations.
K3	=COMPLEX(1,0,"j")	Initial guess for V_2 .
L3	=IMABS(K3)	Magnitude of initial V_2 .
M3	=180*IMARGUMENT(K3)/PI()	Angle of initial V_2 in deg.
N3:P3	Blank cells.	No calculations.
Q3	=COMPLEX(1,0,"j")	Initial value for calculated V_3 .
R3	=COMPLEX(1,0,"j")	Initial value for corrected V_3 ; magnitude must be constant.
S3	=IMABS(R3)	Magnitude of initial corrected V_3 .
T3	=180*IMARGUMENT(Q3)/PI()	Angle of initial calculated V_3 in deg.

SPREADSHEET IMPLEMENTATIONS FOR SOLVING POWER-FLOW PROBLEMS

Table 5 (continued): Microsoft Excel formulas for implementing the Gauss-Seidel method.

Cell	Formula	Comments
I4	=IMDIV(COMPLEX(\$E\$10,-\$F\$10,"j"), IMCONJUGATE(K3))	The term $\frac{P_{2,sch} - jQ_{2,sch}}{V_2^{(m-1)*}}$ of Equation (14).
J4	=IMSUM(IMPRODUCT(\$B\$4,\$C\$9), IMPRODUCT(\$D\$4,R3))	The term $Y_{21} V_1 + Y_{23} V_3^{(m-1)}$ of Equation (14).
K4	=IMDIV(IMSUB(I4,J4),\$C\$4)	Completes the calculation of $V_2^{(m)}$ in Equation (14) using intermediate results in cells I4 and J4.
L4	=IMABS(K4)	Magnitude of updated V_2 .
M4	=180*IMARGUMENT(K4)/PI()	Angle of updated V_2 in deg.
N4	=IMSUM(IMPRODUCT(\$B\$5,\$C\$9), IMSUM(IMPRODUCT(\$C\$5,K4), IMPRODUCT(\$D\$5,R3)))	The term $\sum_{k=1}^N Y_{3k} V_k$ of Equation (6).
O4	=IMDIV(COMPLEX(\$E\$11,-P4,"j"), IMCONJUGATE(R3))	The term $\frac{P_{3,sch} - jQ_{3,sch}}{V_3^{(m-1)*}}$ of Equation (14).
P4	=-IMAGINARY(IMPRODUCT(IMCONJUGATE(R3),N4))	Q_3 computed as the negative of the imaginary part of $V_3^* \sum_{k=1}^N Y_{3k} V_k$ as indicated in Equation (6).
Q4	=IMDIV(IMSUM(IMSUB(O4,N4), IMPRODUCT(\$D\$5,R3)),\$D\$5)	Updated value $V_3^{(m)}$ calculated from Equation (14). Observe that in the cell formula $Y_{33} V_3$ was added back since this term is not present explicitly in Equation (14). The reason is that cell N4 uses $Y_{33} V_3$ when computing Q_3 according to Equation (6), but this term has to be excluded when computing V_3 .
R4	=COMPLEX(S4*COS(IMARGUMENT(Q4)),S4*SIN(IMARGUMENT(Q4)),"j")	Corrects $V_3^{(m)}$ to match the voltage magnitude at bus 3. The angle (argument) is the same as in cell Q4.
S4	=S3	Copies the constant voltage magnitude at bus 3.
T4	=180*IMARGUMENT(Q4)/PI()	Angle of updated V_3 in deg.

Table 5 (continued): Microsoft Excel formulas for implementing the Gauss-Seidel method.

Cell	Formula	Comments
I5:T11	Copies of cell range I4:T4.	Replicates cell formulas in rows below until convergence is achieved.
V2:AC2	Labels for various quantities.	Final calculations section.
V3	=IMSUM(IMPRODUCT(\$B\$4,\$C\$9), IMSUM(IMPRODUCT(\$C\$4,K11), IMPRODUCT(\$D\$4,R11)))	The term $\sum_{k=1}^N Y_{2k} V_k$ of Equation (6).
W3	=IMSUM(IMPRODUCT(\$B\$5,\$C\$9), IMSUM(IMPRODUCT(\$C\$5,K11), IMPRODUCT(\$D\$5,R11)))	The term $\sum_{k=1}^N Y_{3k} V_k$ of Equation (6).
X3	=IMREAL(IMPRODUCT(IMCONJUGATE(K11),V3))	P_2 computed as the real part of $V_2^* \sum_{k=1}^N Y_{2k} V_k$ according to Equation (6).
Y3	=-IMAGINARY(IMPRODUCT(IMCONJUGATE(K11),V3))	Q_2 computed as the negative of the imaginary part of $V_2^* \sum_{k=1}^N Y_{2k} V_k$ according to Equation (6).
Z3	=IMREAL(IMPRODUCT(IMCONJUGATE(R11),W3))	P_3 computed as the real part of $V_3^* \sum_{k=1}^N Y_{3k} V_k$ according to Equation (6).
AA3	=-IMAGINARY(IMPRODUCT(IMCONJUGATE(R11),W3))	Q_3 computed as the negative of the imaginary part of $V_3^* \sum_{k=1}^N Y_{3k} V_k$ according to Equation (6).
AB3	=-X3-Z3	Real power P_1 at slack bus (balance of real power).
AC3	=-Y3-AA3	Reactive power Q_1 at slack bus (balance of reactive power).

5. Power-flow solution by the Newton-Raphson method

5.1 The Newton-Raphson method

To set up the Newton-Raphson numerical method, we employ the power-flow expressions given by Equations (7) and (8). These equations are more convenient than Equation (6) during the computation of the Jacobian matrix, as we will show later.

For simplicity of discussion, we will assume that we have N buses and that all buses, except the slack bus ($i = 1$), are load buses with prescribed demands P_{di} and Q_{di} . Denoting the specified values $|V_1|$ and δ_1 for the slack bus, then each of the remaining buses in the network has the two state variables $|V_i|$ and δ_i to be determined by the power-flow solution. The objective of the Newton-Raphson method is to produce values for $|V_i|$ and δ_i that will match the prescribed P_{di} and Q_{di} as determined from Equations (7) and (8).

At each iteration of the method, new estimates of $|V_i|$ and δ_i for the non-slack buses ($i = 2, 3, \dots, N$) are generated. At the end of each iteration, the power mismatch is given by

$$\Delta P_i = P_{i,sch} - P_i, \quad (15)$$

$$\Delta Q_i = Q_{i,sch} - Q_i. \quad (16)$$

To understand the formulation of the method, we consider a three-bus system. Bus 1 is the slack bus, while buses 2 and 3 are load buses. The basic idea is to view the power mismatches at the load buses as first-order variations resulting from a Taylor series expansion. In this case, the mismatch equations for the real power at the load buses ($i = 2, 3$) are

$$\Delta P_i = \frac{\partial P_i}{\partial |V_2|} \Delta |V_2| + \frac{\partial P_i}{\partial |V_3|} \Delta |V_3| + \frac{\partial P_i}{\partial \delta_2} \Delta \delta_2 + \frac{\partial P_i}{\partial \delta_3} \Delta \delta_3. \quad (17)$$

Similarly, the reactive power mismatch for $i = 2, 3$ are

$$\Delta Q_i = \frac{\partial Q_i}{\partial |V_2|} \Delta |V_2| + \frac{\partial Q_i}{\partial |V_3|} \Delta |V_3| + \frac{\partial Q_i}{\partial \delta_2} \Delta \delta_2 + \frac{\partial Q_i}{\partial \delta_3} \Delta \delta_3. \quad (18)$$

Equations (17) and (18) can be written in matrix form,

$$\begin{bmatrix} \Delta P_2 \\ \Delta P_3 \\ \Delta Q_2 \\ \Delta Q_3 \end{bmatrix} = \begin{bmatrix} \frac{\partial P_2}{\partial |V_2|} & \frac{\partial P_2}{\partial |V_3|} & \frac{\partial P_2}{\partial \delta_2} & \frac{\partial P_2}{\partial \delta_3} \\ \frac{\partial P_3}{\partial |V_2|} & \frac{\partial P_3}{\partial |V_3|} & \frac{\partial P_3}{\partial \delta_2} & \frac{\partial P_3}{\partial \delta_3} \\ \frac{\partial Q_2}{\partial |V_2|} & \frac{\partial Q_2}{\partial |V_3|} & \frac{\partial Q_2}{\partial \delta_2} & \frac{\partial Q_2}{\partial \delta_3} \\ \frac{\partial Q_3}{\partial |V_2|} & \frac{\partial Q_3}{\partial |V_3|} & \frac{\partial Q_3}{\partial \delta_2} & \frac{\partial Q_3}{\partial \delta_3} \end{bmatrix} \begin{bmatrix} \Delta |V_2| \\ \Delta |V_3| \\ \Delta \delta_2 \\ \Delta \delta_3 \end{bmatrix}. \quad (19)$$

The 4x4 matrix in Equation (19) is the *Jacobian* and it will be denoted by **J**. We can summarize the numerical algorithm as follows:

1. Provide initial estimates for $|V_i|^{(0)}$ and $\delta_i^{(0)}$, $i = 2, 3, \dots, N$.
2. Use these estimates to calculate $P_i^{(0)}$, $Q_i^{(0)}$, $\Delta P_i^{(0)}$, $\Delta Q_i^{(0)}$, and the elements of the Jacobian matrix **J**.

3. Solve for the initial corrections $\Delta |V_i|^{(0)}$ and $\Delta \delta_i^{(0)}$ using Equation (19). This step requires matrix inversion of \mathbf{J} .
4. Obtain new estimates

$$|V_i|^{(1)} = |V_i|^{(0)} + \Delta |V_i|^{(0)}, \quad (20)$$

$$\delta_i^{(1)} = \delta_i^{(0)} + \Delta \delta_i^{(0)}. \quad (21)$$

5. Use the new values of $|V_i|^{(1)}$ and $\delta_i^{(1)}$ as starting values for iteration 2 and repeat the above steps. In more general terms, calculate

$$|V_i|^{(m+1)} = |V_i|^{(m)} + \Delta |V_i|^{(m)}, \quad (22)$$

$$\delta_i^{(m+1)} = \delta_i^{(m)} + \Delta \delta_i^{(m)} \quad (23)$$

until a prespecified precision index is satisfied.

We now apply the procedure outlined above to the power system of Figure 1. Since bus 3 is a constant voltage bus and not a load bus, we need to modify the preceding formulation slightly. For this power system, the variables of interest are $|V_2|$, δ_2 , and δ_3 . Taking into account that $|V_1| = 1.0$ p.u., $\delta_1 = 0^\circ$, $|V_3| = 1.0$ p.u., and the \mathbf{Y}_{bus} matrix of Equation (2), we write out the equations that need to be solved in accordance with Equations (7) and (8):

$$\begin{aligned} -2 = & (10)(|V_2|)(1.0)\cos(90^\circ + 0^\circ - \delta_2) + (15)(|V_2|)(|V_2|)\cos(-90^\circ + \delta_2 - \delta_2) \\ & + (5)(|V_2|)(1.0)\cos(90^\circ + \delta_3 - \delta_2), \end{aligned}$$

$$\begin{aligned} -0.5 = & -(10)(|V_2|)(1.0)\sin(90^\circ + 0^\circ - \delta_2) - (15)(|V_2|)(|V_2|)\sin(-90^\circ + \delta_2 - \delta_2) \\ & - (5)(|V_2|)(1.0)\sin(90^\circ + \delta_3 - \delta_2), \end{aligned}$$

$$\begin{aligned} 1 = & (2.5)(1.0)(1.0)\cos(90^\circ + 0^\circ - \delta_3) + (5)(1.0)(|V_2|)\cos(90^\circ + \delta_2 - \delta_3) \\ & + (7.5)(1.0)(1.0)\cos(-90^\circ + \delta_3 - \delta_3). \end{aligned}$$

Simplifying the right-hand sides of the preceding expressions yields

$$P_2 = 10|V_2| \sin(\delta_2) - 5|V_2| \sin(\delta_3 - \delta_2),$$

$$Q_2 = -10|V_2| \cos(\delta_2) + 15|V_2|^2 - 5|V_2| \cos(\delta_3 - \delta_2),$$

$$P_3 = 2.5\sin(\delta_3) - 5|V_2| \sin(\delta_2 - \delta_3).$$

Thus, the elements of the Jacobian matrix are given by

SPREADSHEET IMPLEMENTATIONS FOR SOLVING POWER-FLOW PROBLEMS

$$\frac{\partial P_2}{\partial |V_2|} = 10 \sin(\delta_2) - 5 \sin(\delta_3 - \delta_2),$$

$$\frac{\partial P_2}{\partial \delta_2} = 10 |V_2| \cos(\delta_2) + 5 |V_2| \cos(\delta_3 - \delta_2),$$

$$\frac{\partial P_2}{\partial \delta_3} = -5 |V_2| \cos(\delta_3 - \delta_2),$$

$$\frac{\partial Q_2}{\partial |V_2|} = -10 \cos(\delta_2) + 30 |V_2| - 5 \cos(\delta_3 - \delta_2),$$

$$\frac{\partial Q_2}{\partial \delta_2} = 10 |V_2| \sin(\delta_2) - 5 |V_2| \sin(\delta_3 - \delta_2),$$

$$\frac{\partial Q_2}{\partial \delta_3} = 5 |V_2| \sin(\delta_3 - \delta_2),$$

$$\frac{\partial P_3}{\partial |V_2|} = -5 \sin(\delta_2 - \delta_3),$$

$$\frac{\partial P_3}{\partial \delta_2} = -5 |V_2| \cos(\delta_2 - \delta_3),$$

$$\frac{\partial P_3}{\partial \delta_3} = 2.5 \cos(\delta_3) + 5 |V_2| \cos(\delta_2 - \delta_3).$$

The above partial derivatives form the Jacobian matrix

$$\mathbf{J} = \begin{bmatrix} \frac{\partial P_2}{\partial |V_2|} & \frac{\partial P_2}{\partial \delta_2} & \frac{\partial P_2}{\partial \delta_3} \\ \frac{\partial P_3}{\partial |V_2|} & \frac{\partial P_3}{\partial \delta_2} & \frac{\partial P_3}{\partial \delta_3} \\ \frac{\partial Q_2}{\partial |V_2|} & \frac{\partial Q_2}{\partial \delta_2} & \frac{\partial Q_2}{\partial \delta_3} \end{bmatrix}.$$

For the initial estimates $|V_2| = 1.0$ p.u., $\delta_2 = 0^\circ$, and $\delta_3 = 0^\circ$, the initial Jacobian matrix becomes

$$\mathbf{J}^{(0)} = \begin{bmatrix} 0 & 15 & -5 \\ 0 & -5 & 7.5 \\ 15 & 0 & 0 \end{bmatrix}.$$

The initial power mismatches are

$$\Delta P_2 = P_2 + 2.0 = 0 + 2.0 = 2.0,$$

$$\Delta Q_2 = Q_2 + 0.5 = 0 + 0.5 = 0.5,$$

$$\Delta P_3 = P_3 - 1.0 = 0 - 1.0 = -1.0.$$

Consequently, the corrected values after the first iteration are

$$\begin{bmatrix} |V_2|^{(1)} \\ \delta_2^{(1)} \\ \delta_3^{(1)} \end{bmatrix} = \begin{bmatrix} 1.0 \\ 0 \\ 0 \end{bmatrix} - \begin{bmatrix} 0 & 15 & -5 \\ 0 & -5 & 7.5 \\ 15 & 0 & 0 \end{bmatrix}^{-1} \begin{bmatrix} 2.0 \\ -1.0 \\ 0.5 \end{bmatrix} = \begin{bmatrix} 0.9667 \\ -0.1143 \\ 0.0571 \end{bmatrix}.$$

Although we assumed that $\delta_2 = 0^\circ$, and $\delta_3 = 0^\circ$ for simplicity, these angles must be converted to radians before performing the necessary matrix operations. The procedure is repeated until the variables of interest satisfy a prescribed precision index. Once this is achieved, we use the values of $|V_2|$, δ_2 , and δ_3 given by the algorithm to compute Q_3 from Equation (8). This completes the solution process.

5.2 Spreadsheet implementation of the Newton-Raphson method

In this section we present a spreadsheet implementation of the Newton-Raphson method. Again, we take the power system of Figure 1. The input data section is identical to that of the Gauss-Seidel method discussed in Section 4.2 (see Figures 2 and 3).

The power-flow solution is shown in Figure 5. The calculations reveal that acceptable solutions are attained after 3 iterations for a precision index of 10^{-5} . As expected, the solutions agree with those obtained by the Gauss-Seidel method (see Table 4). In general, the Newton-Raphson method converges to the solution faster than the Gauss-Seidel method. In some instances, ill-conditioned problems lead to divergence by either method.

As can be seen in Figure 5, the Newton-Raphson method covers the cell range H1:T15. A complete list of Microsoft Excel formulas is given in Table 6.

SPREADSHEET IMPLEMENTATIONS FOR SOLVING POWER-FLOW PROBLEMS

	H	I	J	K	L	M	N	O	P	Q	R	S	T
1	Power-flow solutions by Newton Raphson method												
2	Iter.	Var.	Value	Value	Jacobian			Inverse of Jacobian			Power mismatch	Corr.	
3		V ₂	1.0000	1.0000	0.0000	15.0000	-5.0000	0.0000	0.0000	0.0667	dP ₂	2.0000	0.0333
4	0	d ₂	0.000	0.000	0.0000	-5.0000	7.5000	0.0857	0.0571	0.0000	dP ₃	-1.0000	0.1143
5		d ₃	0.000	0.000	15.0000	0.0000	0.0000	0.0571	0.1714	0.0000	dQ ₂	0.5000	-0.0571
6		V ₂	0.9667	0.9667	-1.9933	14.3661	-4.7625	0.0089	-0.0024	0.0721	dP ₂	0.0731	0.0116
7	1	d ₂	-0.1143	-6.548	0.8530	-4.7625	7.2584	0.0901	0.0581	0.0092	dP ₃	-0.0327	0.0061
8		d ₃	0.0571	3.274	14.1385	-1.9269	0.8245	0.0581	0.1761	-0.0024	dQ ₂	0.1506	-0.0019
9		V ₂	0.9551	0.9551	-2.0928	14.1805	-4.6988	0.0096	-0.0025	0.0741	dP ₂	0.0012	0.0002
10	2	d ₂	-0.1204	-6.896	0.8921	-4.6988	7.1944	0.0913	0.0584	0.0101	dP ₃	-0.0005	0.0001
11		d ₃	0.0590	3.382	13.8050	-1.9988	0.8520	0.0584	0.1775	-0.0026	dQ ₂	0.0022	0.0000
12		V ₂	0.9549	0.9549	-2.0944	14.1777	-4.6978	0.0096	-0.0025	0.0741	dP ₂	0.0000	0.0000
13	3	d ₂	-0.1205	-6.902	0.8927	-4.6978	7.1934	0.0913	0.0584	0.0101	dP ₃	0.0000	0.0000
14		d ₃	0.0590	3.383	13.8000	-2.0000	0.8525	0.0584	0.1775	-0.0026	dQ ₂	0.0000	0.0000
15	NOTE:d_i (rad)d_i (deg)												

Figure 5: Spreadsheet implementation of the Newton-Raphson method for determining the power flows in the network of Figure 1.

Table 6: Microsoft Excel formulas for the Newton-Raphson method.

Cell	Formula	Comments
H3:H14	List of numbers from 0 to 3.	Iteration number. Some cells are blank.
H2:T2	Labels for various quantities.	Newton-Raphson section.
I15:K15	Labels to highlight angles in radians and degrees.	No calculations.
I3:I14	Labels for variables of interest.	$ V_2 , \delta_2, \delta_3$.
J3:J14	Numerical values corresponding to variables of interest.	δ_2 and δ_3 in radians.
K3:K14	Numerical values corresponding to variables of interest.	δ_2 and δ_3 in degrees.
L3:N14	Jacobian matrices.	All matrices are 3x3.
O3:Q14	Inverse of Jacobian matrices.	All matrices are 3x3.
R3:R14	Labels for power mismatches.	$\Delta P_2, \Delta P_3, \Delta Q_2$.
S3:S14	Numerical values corresponding to power mismatches.	$\Delta P_2, \Delta P_3, \Delta Q_2$.
T3:T14	Correction terms.	$\Delta V_2 , \Delta \delta_2, \Delta \delta_3$.
J3:J5	Initial estimates for variables of interest.	$ V_2 ^{(0)}, \delta_2^{(0)}, \delta_3^{(0)}$.
K3	=J3	Copies the value of $ V_2 ^{(0)}$.
K4	=J4*180/PI()	Converts $\delta_2^{(0)}$ to degrees.
K5	=J5*180/PI()	Converts $\delta_3^{(0)}$ to degrees.

Table 6 (continued): Microsoft Excel formulas for the Newton-Raphson method.

Cell	Formula	Comments
L3	=10*SIN(J4)-5*SIN(J5-J4)	Jacobian element $\frac{\partial P_2}{\partial V_2 }$.
M3	=10*J3*COS(J4)+5*J3*COS(J5-J4)	Jacobian element $\frac{\partial P_2}{\partial \delta_2}$.
N3	=-5*J3*COS(J5-J4)	Jacobian element $\frac{\partial P_2}{\partial \delta_3}$.
L4	=-5*SIN(J4-J5)	Jacobian element $\frac{\partial P_3}{\partial V_2 }$.
M4	=-5*J3*COS(J4-J5)	Jacobian element $\frac{\partial P_3}{\partial \delta_2}$.
N4	=2.5*COS(J5)+5*J3*COS(J4-J5)	Jacobian element $\frac{\partial P_3}{\partial \delta_3}$.
L5	=-10*COS(J4)+30*J3-5*COS(J5-J4)	Jacobian element $\frac{\partial Q_2}{\partial V_2 }$.
M5	=10*J3*SIN(J4)-5*J3*SIN(J5-J4)	Jacobian element $\frac{\partial Q_2}{\partial \delta_2}$.
N5	=5*J3*SIN(J5-J4)	Jacobian element $\frac{\partial Q_2}{\partial \delta_3}$.
O3:Q5	=MINVERSE(L3:N5)	Inverse of Jacobian matrix in cell range L3:N5.
S3	=10*J3*SIN(J4)-5*J3*SIN(J5-J4)-\$E\$10	Power mismatch ΔP_2 .
S4	=2.5*SIN(J5)-5*J3*SIN(J4-J5)-\$E\$11	Power mismatch ΔP_3 .
S5	=-10*J3*COS(J4)+15*J3^2-5*J3*COS(J5-J4)-\$F\$10	Power mismatch ΔQ_2 .
T3:T5	=MMULT(O3:Q5,S3:S5)	Computes correction terms $\Delta V_2 ^{(0)}$, $\Delta \delta_2^{(0)}$, $\Delta \delta_3^{(0)}$. Multiplies the inverse of the Jacobian in O3:Q5 by the vector of power mismatch in S3:S5.
J6	=J3-T3	New estimate $ V_2 ^{(1)}$.
J7	=J4-T4	New estimate $\delta_2^{(1)}$.
J8	=J5-T5	New estimate $\delta_3^{(1)}$.
K6:T8	Copies of the formulas in cell range K3:T5.	Completes calculations for iteration 1.
J9:T14	Copies of the formulas in cell range J6:T8.	Replicate formulas as needed until solutions are within some precision index.

6. Classroom experience and pedagogical insights

The Gauss-Seidel and the Newton-Raphson methods are among the most popular numerical techniques encountered in power system courses. The authors introduce these techniques to their undergraduate students when discussing the power-flow problem. To reinforce conceptual understanding of these methods, the students are asked to analyze the power flows in a small three-bus system, such as the one given in Figure 1. To facilitate computer implementation, the authors make the spreadsheets described in this paper available to their students. This approach has pedagogical advantages since students are relieved from the burden of learning new software. By simply modifying the necessary spreadsheet cells, the students can easily determine solutions to their assigned problem.

From the authors' experience throughout the years, students express positive attitudes toward the spreadsheet implementation of their power-flow project. Ease of implementation; widespread availability of spreadsheets; convenient tracking and displaying of numerical results; transparency of results that are often obscured by specialized power system analysis programs, are among the most cited comments.

The authors recommend the use of the spreadsheet approach to solve only small power systems, perhaps up to three buses. Large systems are more suitably analyzed using specialized software such as PSS/E or PowerWorld. The spreadsheet approach is effective in other respects such as gaining insight into the numerical techniques, making sense of the convergence or divergence of computer-generated solutions, and developing intuition about well- or ill-conditioned systems.

7. Conclusions

In this paper we have presented spreadsheet implementations of two widely used methods for solving power-flow problems. The Gauss-Seidel and the Newton-Raphson numerical methods are introduced to students in power system analysis courses. Although the mathematical underpinnings are found in courses such as numerical analysis, power systems provide a suitable real-world application upon which constructivist activities can be designed by instructors. Spreadsheets provide students with an easily accessible tool with which mathematical models of real systems can be built and analyzed. Furthermore, spreadsheets lend themselves to answering "what if" questions when quantities such as real power, reactive power, or bus voltage change to new quiescent operating conditions.

References

- [1] Acarnley, P. A power system load flow analysis using an Excel workbook. Online available: <http://www.engsc.ac.uk/downloads/awards/load.pdf>. Accessed: August 13, 2008.
- [2] Bergen, A. R. and Vittal, V. (1999). Power systems analysis, 2nd edition. Prentice Hall.
- [3] Glover, J. D. and Sarma, M. S. (2001). Power system analysis and design, 3rd edition. Thomson-Engineering.
- [4] Grainger, J. and Stevenson, W. D. (1994). Power systems analysis. McGraw-Hill Publishing Co.
- [5] Rao, N. D. and Haddad, N. Y. (1990). Typical applications of new generation spread-

sheets to power system problems, *Proceedings of the 33rd Midwest Symposium on Circuits and Systems*, Vol. 1, 159–162, August 12-14, 1990.

- [6] Zeljković, C. and Gaćanović, M. An example of using Microsoft Excel Solver for power network calculations. Online available: http://phd.etfbl.net/files/Works_PDF/Zelj-kovic%20Cedomir.pdf. Accessed: August 13, 2008.